The SAS® Programmer’s PROC REPORT Handbook

ODS Companion

Jane Eslinger
# Contents

About this Book .............................................................................................................................. vii
About the Author ............................................................................................................................. xi
Preface ............................................................................................................................................. xiii

Chapter 1 Introduction: Overview of Chapters .............................................................................. 1

Chapter 2 PROC TEMPLATE Overview ...................................................................................... 5
  2.1 Introduction .................................................................................................................................. 5
  2.2 Default Style Templates ............................................................................................................. 6
  2.3 Location of Style Templates ....................................................................................................... 7
  2.4 Viewing Style Templates ........................................................................................................... 9
  2.5 General Syntax of PROC TEMPLATE ....................................................................................... 12
  2.6 STYLE versus CLASS Statements .......................................................................................... 14
  2.7 Inheritance ............................................................................................................................... 14
  2.8 Style Attributes ......................................................................................................................... 15
  2.9 Key Takeaways .......................................................................................................................... 16

Chapter 3 Style Elements ........................................................................................................... 17
  3.1 Introduction ............................................................................................................................... 17
  3.2 Style Elements Used by PROC REPORT .................................................................................. 18
  3.3 Setup for Chapter Examples ..................................................................................................... 18
  3.4 Plain Output Generated without Attribute Changes ................................................................ 20
  3.5 Header Style Element Changes ............................................................................................... 24
  3.6 LineContent Style Element Changes ...................................................................................... 28
  3.7 DataEmphasis Style Element Changes .................................................................................... 32
  3.8 Data Style Element Changes ..................................................................................................... 36
  3.9 Table Style Element Changes .................................................................................................. 40
Chapter 4 Style Overrides within PROC REPORT

4.1 Introduction

4.2 STYLE= Default Locations and Order of Precedence

4.3 STYLE= in PROC REPORT Statement

4.4 STYLE= in DEFINE Statement
   4.4.1 Using CELLWIDTH= Style Attribute
   4.4.2 Applying Trafficlighting
   4.4.3 Using Multiple Locations in the Same DEFINE Statement

4.5 STYLE= in BREAK and RBREAK Statements

4.6 STYLE= in COMPUTE Statement

4.7 CALL DEFINE STATEMENT

4.8 Key Takeaways

Chapter 5 PDF Destination

5.1 Introduction

5.2 Style Templates with ODS PDF

5.3 Border Manipulation
   5.3.1 Manipulate Borders Already in the Table
   5.3.2 Insert New Borders
   5.3.3 Border Width Values

5.4 Paging

5.5 PROC REPORT SPANROWS Option

5.6 Accessibility

5.7 Key Takeaways

Chapter 6 RTF and Tagsets.RTF Destinations

6.1 Introduction

6.2 Differences between the RTF and Tagsets.RTF Destinations
   6.2.1 Titles and Footnotes
   6.2.2 CONTINUED Text
   6.2.3 Watermark Text

6.3 Effect of Style Element Changes in Tagsets.RTF
   6.3.1 Header Style Element Changes
   6.3.2 LineContent Style Element Changes
   6.3.3 DataEmphasis Style Element Changes
   6.3.4 Data Element Style Changes
Chapter 8 HTML and HTML5 Destinations ................................................................. 165
  8.1 Introduction ................................................................................................................. 165
  8.2 Style Differences between the HTML and HTML5 Destinations ......................... 166
    8.2.1 Header Style Element Changes ............................................................................... 167
    8.2.2 LineContent Style Element Changes .................................................................. 168
    8.2.3 DataEmphasis Style Element Changes .............................................................. 169
    8.2.4 Data Element Style Changes .............................................................................. 170
    8.2.5 Table Element Style Changes ............................................................................ 170
    8.2.6 Container Element Style Changes .................................................................... 171
  8.3 Borders ......................................................................................................................... 172
  8.4 Hyperlinks .................................................................................................................... 174
  8.5 Images .......................................................................................................................... 175
  8.6 Accessibility .................................................................................................................. 176
  8.7 Key Takeaways ............................................................................................................ 176

Chapter 9 ODS Destination for PowerPoint ............................................................... 177
  9.1 Introduction .................................................................................................................. 177
  9.2 PowerPointLight and PowerPointDark Style Templates .......................................... 178
  9.3 Style Template Changes ............................................................................................. 180
  9.4 Effects of Tables with Large Amounts of Data ......................................................... 187
    9.4.1 Many Columns of Data ...................................................................................... 187
    9.4.2 Many Columns and Rows of Data ................................................................... 192
    9.4.3 Many Header Rows ......................................................................................... 199
  9.5 Inline Formatting Functions ....................................................................................... 203
  9.6 Style Attributes for Images ......................................................................................... 206
  9.7 Key Takeaways ............................................................................................................ 211

ODS References ............................................................................................................ 213

Index ................................................................................................................................. 215
What Does This Book Cover?

The purpose of this book is to describe how PROC REPORT interacts with the major Output Delivery System (ODS) destinations. Each ODS destination behaves differently from the others, and programmers need to be aware that they might have to alter their PROC REPORT code or create a custom style template based on the final destination of their report.

This book covers how PROC REPORT interacts with the following destinations:

- PDF
- RTF
- Tagsets.RTF
- HTML
- HTML5
- ODS destination for Excel
- Tagsets.ExcelXP
- ODS destination for PowerPoint

This book does not discuss how to perform calculations with PROC REPORT or how to structure output. Those topics were covered in *The SAS Programmer’s PROC REPORT Handbook: Basic to Advanced Reporting Techniques* (referred to as *Reporting Techniques* throughout the book).

This book does not cover style templates and PROC TEMPLATE in-depth. Those topics are briefly discussed with regard to how they relate to PROC REPORT.

Is This Book for You?

This book is meant for SAS programmers of all skill levels in all industries who need to create reports. This book will teach you how PROC REPORT works with ODS and the steps that you can take with your code to ensure that you are getting the output in the format that you need and with all the styling enhancements that you want.

What Are the Prerequisites for This Book?

Basic knowledge of SAS programming is necessary in order to understand the concepts and examples in this book. You need a general understanding of the basic concepts of sending output to the ODS destinations.
Scope of This Book

This book is entirely about PROC REPORT and ODS. PROC REPORT is designed specifically for generating reports. Therefore, ODS plays a major role in the final appearance of the report. The book attempts to give an overview about ODS and each destination as it relates to PROC REPORT. A full description of every destination and every style attribute and its uses is beyond the scope of this book.

Although the TEMPLATE procedure is used in this book, it is not covered in depth. The syntax of the TEMPLATE procedure is described only enough to convey the reason the procedure is needed for use with PROC REPORT.

What Should You Know about the Examples?

Software Used to Develop the Book's Content

The examples in this book were developed using Base SAS 9.4TS1M4. The examples can be run in SAS 9.4TS1M0 – SAS 9.4TS1M3, with all available hot fixes installed. The examples that pertain to the ODS destinations for Excel and PowerPoint should be run in the most recent releases, SAS 9.4TS1M4 or SAS 9.4TS1M5.

All examples that are not related to the ODS destinations for Excel and PowerPoint should work in SAS 9.3 as well. But please note that if the example programs are run interactively in SAS 9.3, the NOWD option should be added to the PROC REPORT statement. This book is also compatible with SAS University Edition.

Example Code and Data

Permission to use the Orion Star data, which was used for demonstration purposes throughout this book, was granted by Sean O’Brien and Eric Rossland. All the examples in this book use one SAS data set ORDERS.

ORDERS contains purchase order data. The data set is unique at CUSTOMER_ID, ORDER_ID, and PRODUCT_ID level. A customer can have multiple orders and within each unique order, the customer can purchase multiple products. The QUANTITY variable contains the quantity ordered of a specific product.

To use the example programs from the author page, you need the ORDERS data set in the WORK location. You also need to submit the createfmts.sas program. The program creates two formats, called $cntry and typef, which are used throughout the book. The formats should be stored in a catalog in the WORK location, as well.

You do not have to read this book in its entirety to gain valuable knowledge about PROC REPORT and how it interacts with ODS. Chapters 2, 3, and 4 provide the foundation for the remaining chapters of the book. You might find it helpful to read these chapters first, and then concentrate on the chapter about the destination that is of most interest to you.

You can access the example code and data for this book by linking to its author page at http://support.sas.com/publishing/authors. Select the name of the author. Then, look for the cover thumbnail of this book, and select Example Code and Data to display the SAS programs that are included in this book.

If you are unable to access the code through the website, send email to saspress@sas.com.
SAS University Edition

This book is compatible with SAS University Edition. If you are using SAS University Edition, then begin here: http://support.sas.com/publishing/import_ue.data.html

Output and Graphics

The output in this book is sent to many of the ODS destinations, including PDF, HTML, RTF, Tagsets.ExcelXP, and the Excel destination.

Terminology Used in This Book

The words group, order, and across are capitalized when the context refers to the usage value in the DEFINE statement. All statement names, options, and variable names are capitalized in text.

We Want to Hear from You

SAS Press books are written by SAS Users for SAS Users. We welcome your participation in their development and your feedback on SAS Press books that you are using. Please visit sas.com/books to do the following:

- Sign up to review a book.
- Recommend a topic.
- Request information about how to become a SAS Press author.
- Provide feedback on a book.

Do you have questions about a SAS Press book that you are reading? Contact the author through saspress@sas.com or https://support.sas.com/author_feed back.

SAS has many resources to help you find answers and expand your knowledge. If you need additional help, see our list of resources: sas.com/books.
x The SAS Programmer’s PROC REPORT Handbook: ODS Companion
About the Author

Jane Eslinger is a Senior Technical Support Analyst at SAS world headquarters in Cary, North Carolina. Jane is a SAS Certified Advanced Programmer for SAS(R)9 and a SAS Certified Advanced Visual Business Analyst. She frequently presents PROC REPORT and ODS topics at conferences and SAS users groups. She is the author of numerous papers and *The SAS® Programmer's PROC REPORT Handbook: Basic to Advanced Reporting Techniques*.

In her day-to-day work, Jane enjoys supporting SAS customers using ODS and Base SAS procedures, with an emphasis on PROC REPORT. Prior to joining SAS, Jane served as a statistician and statistical programmer in the social science and clinical research fields. She has a B.S. in Statistics from North Carolina State University.

Learn more about this author by visiting her author page at [http://support.sas.com/Eslinger](http://support.sas.com/Eslinger). There you can download free book excerpts, access example code and data, read the latest reviews, get updates, and more.
9.1 Introduction

In Chapter 3, the style templates that were created did not have a parent template explicitly set in their definition. The results provided a way to determine which style elements affected which part of the table, without complicated inheritance rules that could cause confusion. SAS ships two style templates, Styles.PowerPointLight and Styles.PowerPointDark, one of which has to be used as a parent with the ODS destination for PowerPoint: Styles.PowerPointLight and Styles.PowerPointDark. PowerPointLight is the default style template for the destination. Results from a style template that does not use one of the shipped templates as the parent are unpredictable and are not supported by SAS.

When generating PowerPoint presentations and making style changes to PROC REPORT tables, it is important to remember that they are designed for display on a large screen. To enhance readability, the table will have a larger font and more space between values than any other destination.

This chapter focuses on the PowerPoint destination and how it might force you to alter your PROC REPORT code. The first section of the chapter looks at the style templates used by the destination. Examples demonstrate the effects of making custom changes to those templates. The chapter also covers how to alter the PROC
REPORT code to make sure that the tables can fit on a PowerPoint slide in a coherent, readable form. Finally, the last sections explore in-line formatting options and images.

### 9.2 PowerPointLight and PowerPointDark Style Templates

Before making alterations to the style template used for PowerPoint presentations, it is important to see what the table looks like by default. Example 9.1 creates the same output table as the examples in Chapter 3. However, the results are created using the PowerPointLight style template. The ODS POWERPOINT statement does not specify the STYLE= option because PowerPointLight is the default style template that is used by the destination. Output 9.1 shows the appearance of the default template.

Note that the examples in Sections 9.2 and 9.3 are modified slightly from what was used in Chapter 3. The RBREAK statement has been removed, and well as the compute after block with a LINE statement.

**Example 9.1 Using the PowerPointLight Style Template**

```sas
title;
ods powerpoint file='Output9-1.pptx';
proc report data=orders(obs=20)
        style(report)=
                [pretext='this is pretext'];
        column
                ("Customer Info" customer_country customer_age_group)
                ("Order Info" order_type quantity);
        define customer_country / group format=$cntry.;
        define customer_age_group / group;
        define order_type / group format=typef.;
        compute before customer_country;
               line 'a line statement for ' customer_country $cntry.;
        endcomp;
        break after customer_country / summarize;
run;
ods powerpoint close;
```
The PowerPointDark style template was developed to provide a high contrast between foreground and background colors. The slide color is black, and the text is white. Example 9.2 specifies the PowerPointDark style template in the `STYLE=` option in the ODS POWERPOINT statement. Output 9.2 shows the results.

### Example 9.2 Using the PowerPointDark Style Template

```sql
Title;
ods powerpoint file='Output9-2.pptx' style=styles.PowerPointDark;
proc report data=orders(obs=20)
    style(report)=[pretext='this is pretext'];

    column ("Customer Info" customer_country customer_age_group)
        ("Order Info" order_type quantity);

    define customer_country / group format=$cntry.;
    define customer_age_group / group;
    define order_type / group format=typef.;

    compute before customer_country;
        line 'a line statement for ' customer_country $cntry.;
    endcomp;
    break after customer_country / summarize;
run;
ods powerpoint close;
```

---

<table>
<thead>
<tr>
<th>Customer Country</th>
<th>Customer Age Group</th>
<th>Order Type</th>
<th>Quantity Ordered</th>
</tr>
</thead>
<tbody>
<tr>
<td>Australia</td>
<td>31-45 years</td>
<td>Internet Sale</td>
<td>1</td>
</tr>
<tr>
<td></td>
<td></td>
<td>Retail Sale</td>
<td>4</td>
</tr>
<tr>
<td></td>
<td>46-60 years</td>
<td>Internet Sale</td>
<td>3</td>
</tr>
<tr>
<td>Australia</td>
<td></td>
<td></td>
<td>8</td>
</tr>
<tr>
<td>United States</td>
<td>15-30 years</td>
<td>Internet Sale</td>
<td>2</td>
</tr>
<tr>
<td></td>
<td></td>
<td>Retail Sale</td>
<td>11</td>
</tr>
<tr>
<td></td>
<td>31-45 years</td>
<td>Retail Sale</td>
<td>7</td>
</tr>
<tr>
<td></td>
<td>61-75 years</td>
<td>Retail Sale</td>
<td>1</td>
</tr>
<tr>
<td>United States</td>
<td></td>
<td></td>
<td>21</td>
</tr>
</tbody>
</table>

The `this is pretext` column is included to demonstrate the use of the `STYLE=` option and the `pretext` attribute to set the text color.
Output 9.2 PowerPointDark Output

<table>
<thead>
<tr>
<th>Customer Info</th>
<th>Order Info</th>
</tr>
</thead>
<tbody>
<tr>
<td>Customer Country</td>
<td>Order Type</td>
</tr>
<tr>
<td>Customer Age Group</td>
<td>Quantity Ordered</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Australia</th>
<th>31-45 years</th>
<th>Internet Sale</th>
<th>1</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td></td>
<td>Retail Sale</td>
<td>4</td>
</tr>
<tr>
<td></td>
<td>46-60 years</td>
<td>Internet Sale</td>
<td>3</td>
</tr>
<tr>
<td>Australia</td>
<td></td>
<td></td>
<td>8</td>
</tr>
<tr>
<td>United States</td>
<td>15-30 years</td>
<td>Internet Sale</td>
<td>2</td>
</tr>
<tr>
<td></td>
<td></td>
<td>Retail Sale</td>
<td>11</td>
</tr>
<tr>
<td></td>
<td>31-45 years</td>
<td>Retail Sale</td>
<td>7</td>
</tr>
<tr>
<td></td>
<td>61-75 years</td>
<td>Retail Sale</td>
<td>1</td>
</tr>
<tr>
<td>United States</td>
<td></td>
<td></td>
<td>21</td>
</tr>
</tbody>
</table>

Note that the ODS destination for PowerPoint cannot toggle between style templates within one presentation.

9.3 Style Template Changes

Now that you know what the style templates that are supplied by SAS look like, the next step is to consider how changing the template affects the tables produced by PROC REPORT.

Recall from Chapter 3 that the border attribute changes in the Header style element were not applied (Output 3.2.5). This was a consequence of not using a parent style template. The teal borders are applied, along with the gray background and navy foreground, when the custom template inherits from PowerPointLight. Example 9.3 demonstrates how to inherit from PowerPointLight and change the Header element. Output 9.3 shows the resulting PowerPoint slide.
Example 9.3 Header Style Element Changes

proc template;
define style _ppheader;
parent=styles.powerpointlight;
class header /
   fontfamily='Albany AMT'
color=#264666
fontweight=bold
fontstyle=italic
backgroundcolor=#F0F0F0
borderwidth=2pt
bordercolor=#00929F
borderstyle=solid;
end;
run;

title;
ods powerpoint file='Output9-3.pptx' style=_ppheader;
proc report data=orders(obs=20)
   style(report)=[pretext='this is pretext']
   style(column)=[width=2in];

   column ("Customer Info" customer_country customer_age_group)
        ("Order Info" order_type quantity);

   define customer_country / group format=$cntry.;
define customer_age_group / group;
define order_type / group format=typef.;

   compute before customer_country;
      line 'a line statement for ' customer_country $cntry.;
   endcomp;

   break after customer_country / summarize;
run;
ods powerpoint close;
Output 9.3 Header Element Changes

The header in Output 9.3 has a gray background much like the background color in Output 9.1. Even though the shades of gray are similar, they are actually different. The lighter color in Output 9.3 is represented as #F0F0F0, which has been used throughout this book to show the results of modifying the style template.

Output 9.3 also demonstrates that the borders within the table can be controlled. For many PROC REPORT tables, controlling the borders is an important aspect of the final appearance of the table. The default, Output 9.1, has light gray lines that surround each cell in the table. The FRAME= and RULES= attributes are often used to change the outside and inside borders, respectively.

Example 9.4 sets the FRAME= and RULES= attributes inside the Table element. The goal is to have a teal border around the outside of the table and a teal border between the headers and data. The Table element has the most predictable effect for changing the FRAME= and RULES= attributes for PowerPoint.

Example 9.4 FRAME= and RULES= in the Table Element

```sas
proc template;
define style _ppborder1;
parent=styles.powerpointlight;
  class table /
    frame=box
    rules=groups
    borderwidth=2pt
    bordercolor=#00929F;
  end;
run;
```
title;
ods powerpoint file="Output9-4.pptx" style=ppborder1;
proc report data=orders(obs=20)
    style(report)=[pretext='this is pretext']
    style(header)=[backgroundcolor=white];

column ("Customer Info" customer_country customer_age_group)
    ("Order Info" order_type quantity);

define customer_country / group format=$cntry.;
define customer_age_group / group;
define order_type / group format=typef.;

compute before customer_country;
    line 'a line statement for ' customer_country $cntry.;
endcomp;

break after customer_country / summarize;
run;
ods powerpoint close;

Output 9.4 FRAME= and RULES= in the Table Element

<table>
<thead>
<tr>
<th>Customer Info</th>
<th>Order Info</th>
</tr>
</thead>
<tbody>
<tr>
<td>Customer Country</td>
<td>Customer Age Group</td>
</tr>
<tr>
<td>Australia</td>
<td>31-45 years</td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td>Australia</td>
<td>46-60 years</td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td>United States</td>
<td>15-30 years</td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td>31-45 years</td>
</tr>
<tr>
<td></td>
<td>61-75 years</td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
</tbody>
</table>
The Table element came close to providing the desired output. A teal border was successfully placed around the outside of the table and between the header section of the table and data. The problem with the slide shown in Output 9.4 is that light gray lines can still be seen (not as many as in Output 9.1, but some remain).

Example 9.5 removes the remaining gray lines by using the BORDERSTYLE= attribute inside the Data style element. The results are shown in Output 9.5.

**Example 9.5 Creating Only the Desired Borders**

```sas
proc template;
  define style _ppborder2;
    parent=styles.powerpointlight;
    class table /
      frame=box
      rules=groups
      borderwidth=2pt
      bordercolor=#00929F;
    class data /
      borderstyle=hidden;
    class header /
      borderstyle=hidden;
  end;
run;

title;
ods powerpoint file="Output9-5.pptx" style=_ppborder2;
proc report data=orders(obs=20)
  style(report)=[pretext='this is pretext']
  style(header)=[backgroundcolor=white] ;
  column ("Customer Info" customer_country customer_age_group)
    ("Order Info" order_type quantity);
  define customer_country / group format=$cntry.;
  define customer_age_group / group;
  define order_type / group format=typef.;
  compute before customer_country;
    line 'a line statement for ' customer_country $cntry.;
  endcomp;
  break after customer_country / summarize;
run;
ods powerpoint close;
```
Producing your desired borders might require a bit of trial and error work. The only way to determine whether your code will produce the needed borders is to test it. If the borders cannot be created with a custom style template, then overrides within PROC REPORT can be used to generate the borders.

All the gray lines can be removed with style overrides inside the PROC REPORT code. However, it is likely that you want the gray lines to be removed from all tables that will be included in the presentation. Therefore, the most efficient way to remove the borders is by altering the style template. Example 9.6 uses the BORDERSTYLE= attribute to remove the gray lines from the five style elements used by PROC REPORT.

**Example 9.6 Removing Gray Lines with the Style Template**

```plaintext
proc template;
define style _ppnogray;
parent=styles.powerpointlight;
   class header, data, dataemphasis, linecontent, table /
      borderstyle=hidden;
end;
run;
title;
```
ods powerpoint file="Output9-6.pptx" style=_ppnogray;
proc report data=orders(obs=20)
   style(report)=pretext='this is pretext'
   style(header)=backgroundcolor=white ;
   column ("Customer Info" customer_country customer_age_group)
   ("Order Info" order_type quantity);
   define customer_country / group format=$cntry.;
   define customer_age_group / group;
   define order_type / group format=typef.;
   compute before customer_country;
      line 'a line statement for ' customer_country $cntry.;
   endcomp;
   break after customer_country / summarize;
run;
ods powerpoint close;

Output 9.6 Gray Lines Removed from the Table

<table>
<thead>
<tr>
<th>Customer Country</th>
<th>Customer Age Group</th>
<th>Order Type</th>
<th>Quantity Ordered</th>
</tr>
</thead>
<tbody>
<tr>
<td>Australia</td>
<td>31-45 years</td>
<td>Internet Sale</td>
<td>1</td>
</tr>
<tr>
<td></td>
<td></td>
<td>Retail Sale</td>
<td>4</td>
</tr>
<tr>
<td></td>
<td>46-60 years</td>
<td>Internet Sale</td>
<td>3</td>
</tr>
<tr>
<td>Australia</td>
<td></td>
<td></td>
<td>8</td>
</tr>
<tr>
<td>United States</td>
<td>15-30 years</td>
<td>Internet Sale</td>
<td>2</td>
</tr>
<tr>
<td></td>
<td></td>
<td>Retail Sale</td>
<td>11</td>
</tr>
<tr>
<td></td>
<td>31-45 years</td>
<td>Retail Sale</td>
<td>7</td>
</tr>
<tr>
<td></td>
<td>61-75 years</td>
<td>Retail Sale</td>
<td>1</td>
</tr>
<tr>
<td>United States</td>
<td></td>
<td></td>
<td>21</td>
</tr>
</tbody>
</table>
The ODS destination for PowerPoint substitutes the SAS fonts Albany AMT, Cumberland AMT, and Thorndale AMT fonts for the Windows fonts Arial, Courier New, and Times New Roman, respectively.

9.4 Effects of Tables with Large Amounts of Data

The most difficult aspect of learning to develop PROC REPORT code for the PowerPoint destination is dealing with the change in scale. Again, PowerPoint presentations are meant to be viewed on a large screen. Although PowerPoint presentations are not designed for printing, they are often printed out as a convenience for later reading. The amount of data that can fit on one slide is not nearly the amount of data that can fit on one page in a PDF document. This section discusses PROC REPORT techniques that you can use to get a table to fit well in PowerPoint.

9.4.1 Many Columns of Data

The table in Outputs 9.1 and 9.2 has one row of pretext data, two header rows, and eleven rows of data. The table also contains four columns. The table occupies the entire slide, from top to bottom. The 11 rows of data just barely fit on the slide due to the font size, spacing, and other attributes of the table. The entire table that was used in previous chapters, like Chapter 3, did not fit on one slide. The RBREAK summary row along with the final LINE statement text had to be removed in order to get the table to fit on one slide. Even fewer rows of data would have fit on the slide if the code had included a title or footnote. Including the RBREAK and LINE statement forces the last row of data and the last row of text to flow to a second slide. Adding both a title and footnote forces four rows of data and the row of text to flow to a second slide.

In addition, Examples 9.1 and 9.2 do not specify any column widths for the data. The destination determines the width of the columns based on the formatted values of the variables along with the length of the header text. For this table, none of the headers or data wrapped within their cells because there was enough space on the slide.

Example 9.7 includes a fifth column, TOTAL_RETAIL_PRICE. The goal of this example is to demonstrate how adding another one column of data changes how the table fits on the slide. First, as shown in Output 9.7, the table now extends to a second slide. Second, all of the headers wrap within their cells. Third, the values of the ORDER_TYPE column wrap within their cells. Output 9.7 highlights the ORDER_TYPE column and the new column, TOTAL_RETAIL_PRICE.

**Example 9.7 Adding Another Column to the Table**

title;
ods powerpoint file="Output9-7.pptx";
proc report data=orders(obs=20)
   style(report)=[pretext='this is pretext']
   style(header)=[backgroundcolor=#F0F0F0];
   column ("Customer Info" customer_country customer_age_group)
       ("Order Info" order_type quantity total_retail_price);
   define customer_country / group format=$cntry.;
   define customer_age_group / group;
   define order_type / group format=typef. style(header)
       column=[backgroundcolor=#CCDF9C];
   define total_retail_price / style(header column)=[backgroundcolor=#68C9D2];
compute before customer_country;
    line 'a line statement for ' customer_country $cntry.;
endcomp;

break after customer_country / summarize;
run;
ods powerpoint close;

**Output 9.7 Table with Five Columns**

```
<table>
<thead>
<tr>
<th>Customer Info</th>
<th>Order Info</th>
<th>Total Retail Price for This Product</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Customer Age Group</td>
<td>Order Type</td>
</tr>
<tr>
<td>Customer Country</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Australia</td>
<td>31-45 years</td>
<td>Internet Sale</td>
</tr>
<tr>
<td></td>
<td></td>
<td>Retail Sale</td>
</tr>
<tr>
<td></td>
<td>46-60 years</td>
<td>Internet Sale</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>Australia</td>
<td>15-30 years</td>
<td>Internet Sale</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
```

this is pretext
Output 9.7 shows that just one small change (such as increasing from four columns to five columns) can produce big consequences. Admittedly, the table could be made smaller so that it fits on one slide, without compromising readability. Changes could be made with the style template as demonstrated earlier in this chapter. However, template changes affect every table that is sent to the destination. For the ODS destination for PowerPoint, it is important to consider all of the output that is being generated. Do all tables need to be altered the same way, regardless of whether they are created by PROC REPORT, PROC TABULATE, PROC FREQ, or another procedure? If so, it makes sense to create a custom style template. If only one table within the presentation is not fitting as desired, it is better to make the changes within that one procedure.

There is no right way to make sure that a table fits on a slide. Each table is different, but some of the same techniques can be used for each table. For example, the font size in Output 9.7 is 20pt. The font size can be reduced slightly to allow the table to fit. Example 9.8 demonstrates changing the font size for the table to 18pt. It also makes one other critical change. It changes the label of the TOTAL RETAIL PRICE variable. Output 9.8 shows the table that is now contained on one slide.

**Example 9.8 Creating a Table with 18pt Font Size**

title;
ods powerpoint file="Output9-8.pptx";
proc report data=orders(obs=20)
    style(report)=[pretext='this is pretext']
    style(header)=[backgroundcolor=#F0F0F0]
    style(report column header summary lines)=[fontsize=18pt];
column ("Customer Info" customer_country customer_age_group)  
("Order Info" order_type quantity total_retail_price);

define customer_country / group format=$cntry.;
define customer_age_group / group;
define order_type / group format=typef. style(header
column)=[backgroundcolor=#CCDF9C];
define total_retail_price / 'Total/Retail Price' style(header
column)=[backgroundcolor=#68C9D2];

compute before customer_country;
    line 'a line statement for ' customer_country $cntry.;
endcomp;
break after customer_country / summarize;
run;
ods powerpoint close;

Output 9.8 Table with 18pt Font Size

<table>
<thead>
<tr>
<th>Customer Info</th>
<th>Order Info</th>
</tr>
</thead>
<tbody>
<tr>
<td>Customer Country</td>
<td>Order Type</td>
</tr>
<tr>
<td>Australia</td>
<td>Internet Sale</td>
</tr>
<tr>
<td>Australia</td>
<td>Retail Sale</td>
</tr>
<tr>
<td>Australia</td>
<td>Internet Sale</td>
</tr>
<tr>
<td>Australia</td>
<td>8</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th></th>
<th>Quantity Ordered</th>
<th>Total Retail Price</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>1</td>
<td>$21.80</td>
</tr>
<tr>
<td></td>
<td>4</td>
<td>$223.80</td>
</tr>
<tr>
<td></td>
<td>3</td>
<td>$37.80</td>
</tr>
<tr>
<td></td>
<td>8</td>
<td>$283.40</td>
</tr>
</tbody>
</table>

Notice that the ORDER_TYPE values no long wrap within their cells. This is a direct result of changing the label for the TOTAL_RETAIL_PRICE variable. The shorter label means that the column does not have to be as wide, leaving more room for the other columns.
The font size of the text that was generated with the PRETEXT= attribute did not change. In other destinations, the report location can be used to change the attributes for PRETEXT=. For the ODS destination for PowerPoint, the attributes for the text cannot be controlled through changes in PROC REPORT. The changes must be made through a style template.

Example 9.9 changes the font size attribute in the Note element, which controls the PRETEXT= attribute in PowerPoint. The table with the smaller font size is shown in Output 9.9.

**Example 9.9 Changing the Pretext Font Size**

```plaintext
proc template;
define style _pretext;
parent = styles.powerpointlight;
   class Note /
      fontsize = 18pt
      fontfamily='Albany AMT'
      color=#8E2F8A
      fontweight=bold;
end;
run;

title;
ods powerpoint file="Output9-9.pptx" style=_pretext;
proc report data=orders(obs=20)
   style(report)=[pretext='this is pretext']
   style(header)=[backgroundcolor=#F0F0F0]
   style(report column header summary lines)=[fontsize=18pt];
   column ("Customer Info" customer_country customer_age_group)
   ("Order Info" order_type quantity total_retail_price);

define customer_country / group format=$cntry.;
define customer_age_group / group;
define order_type / group format=typef.;
define total_retail_price / 'Total/Retail Price';

compute before customer_country;
   line 'a line statement for ' customer_country $cntry.;
endcomp;

break after customer_country / summarize;
run;
ods powerpoint close;
```
Output 9.9 Table with 18pt Font Size for Pretext

<table>
<thead>
<tr>
<th>Customer Info</th>
<th>Order Info</th>
</tr>
</thead>
<tbody>
<tr>
<td>Customer Country</td>
<td>Customer Age Group</td>
</tr>
<tr>
<td>Australia</td>
<td>31-45 years</td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td>46-60 years</td>
</tr>
<tr>
<td><strong>Australia</strong></td>
<td></td>
</tr>
<tr>
<td>a line statement for Australia</td>
<td></td>
</tr>
<tr>
<td></td>
<td>United States</td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td>31-45 years</td>
</tr>
<tr>
<td></td>
<td>61-75 years</td>
</tr>
<tr>
<td><strong>United States</strong></td>
<td></td>
</tr>
<tr>
<td>a line statement for United States</td>
<td></td>
</tr>
</tbody>
</table>

9.4.2 Many Columns and Rows of Data

The ODS destination for PowerPoint automatically breaks a table into multiple panels when it is too long or too wide to fit on a single slide. For long tables, a subset of rows is placed on another slide. For wide tables, the destination creates subsets of columns, and each subset is placed on its own slide.

Suppose that one of the required tables is both long and wide. Example 9.10 produces a table that has 7 columns and 66 rows. A PDF file fits more data per page, but even in PDF, 66 rows might not fit on one page. Even though Output 9.10 shows only the first two slides that are created by Example 9.10 the entire table actually spans 21 slides!

Example 9.10 Creating a Long and Wide Table

title;
ods powerpoint file="Output9-10.pptx";
proc report data=orders
  style(header)=[backgroundcolor=#3D5AAE color=white];
  column ("Customer Info" customer_country customer_group customer_age_group)
  ("Order Info" order_type costprice_per_unit quantity total_retail_price);
define customer_country / group format=$cntry.
    style(column)=[backgroundcolor=#F0F0F0];
define customer_group / group style(column)=[backgroundcolor=#F0F0F0];
define customer_age_group / group style(column)=[backgroundcolor=#F0F0F0];
define order_type / group format=typef.
    style(column)=[backgroundcolor=#CCDF9C];
define costprice_per_unit / style(column)=[backgroundcolor=#CCDF9C];
define quantity / style(column)=[backgroundcolor=#CCDF9C];
define total_retail_price / style(column)=[backgroundcolor=#CCDF9C];

    break after customer_country / summarize;
run;
ods powerpoint close;

**Output 9.10 Slides for the Long and Wide Table**

<table>
<thead>
<tr>
<th>Customer Country</th>
<th>Customer Group Name</th>
<th>Customer Age Group</th>
</tr>
</thead>
<tbody>
<tr>
<td>Australia</td>
<td>Internet/Catalog Customers</td>
<td>61-75 years</td>
</tr>
<tr>
<td></td>
<td>Orion Club Gold members</td>
<td>46-60 years</td>
</tr>
<tr>
<td></td>
<td>Orion Club members</td>
<td>31-45 years</td>
</tr>
<tr>
<td></td>
<td></td>
<td>46-60 years</td>
</tr>
</tbody>
</table>
The entire table must be included in the presentation, but it needs to be more readable. The table must span fewer slides to make the presentation more useful. Here are some constructive criticisms of Output 9.10:

- All the columns should stay together on one slide.
- The Total Retail Price column is too wide.
- The Total Retail Price column header is too wordy.
- The information for a country should be kept together and not span across slides.

Example 9.11 demonstrates the techniques for altering a table that is both long and wide so that it fits better in the presentation. Output 9.11 shows the first slide of the resulting presentation.

Example 9.11 Creating a Long and Wide Table

title;
ods powerpoint file="Output9-11.pptx";
proc report data=orders
  style(header)=[backgroundcolor=#3D5AAE color=white]
  style(report column header summary)=[fontsize=16pt];
  column ("Customer Info" customer_country customer_group customer_age_group)
  ("Order Info" order_type costprice_per_unit quantity total_retail_price);

<table>
<thead>
<tr>
<th>Order Info</th>
<th>Cost Price Per Unit</th>
<th>Quantity Ordered</th>
<th>Total Retail Price for This Product</th>
</tr>
</thead>
<tbody>
<tr>
<td>Catalog Sale</td>
<td>$429.60</td>
<td>13</td>
<td>$1,061.30</td>
</tr>
<tr>
<td>Retail Sale</td>
<td>$552.05</td>
<td>24</td>
<td>$1,545.50</td>
</tr>
<tr>
<td>Catalog Sale</td>
<td>$131.15</td>
<td>12</td>
<td>$515.20</td>
</tr>
<tr>
<td>Internet Sale</td>
<td>$132.60</td>
<td>8</td>
<td>$395.70</td>
</tr>
<tr>
<td>Retail Sale</td>
<td>$1,616.05</td>
<td>84</td>
<td>$5,696.69</td>
</tr>
<tr>
<td>Catalog Sale</td>
<td>$46.90</td>
<td>1</td>
<td>$102.90</td>
</tr>
<tr>
<td>Internet Sale</td>
<td>$44.00</td>
<td>5</td>
<td>$218.20</td>
</tr>
</tbody>
</table>
define customer_country / group format=$cntry. 'Country'
  style(column)=[backgroundcolor=#F0F0F0];
define customer_group / group 'Group' style(column)=[backgroundcolor=#F0F0F0
  width=1.5in];
define customer_age_group / group 'Age Group'
  style(column)=[backgroundcolor=#F0F0F0];
define order_type / group format=typef. style(column)=[backgroundcolor=#CCDF9C
  width=1.3in];
define costprice_per_unit / 'Cost Price'
  style(column)=[backgroundcolor=#CCDF9C width=1.2in];
define quantity / style(column)=[backgroundcolor=#CCDF9C width=1.1in];
define total_retail_price / 'Total Retail Price'
  style(column)=[backgroundcolor=#CCDF9C width=1.2in];
break after customer_country / summarize page;
run;
ods powerpoint close;

1. The font size is reduced to 16pt. Although this change is made in the PROC REPORT statement, it was not
   the first change made to get the table to fit better on the slides. Changing the font size was necessary but it
   should be considered only after other attempts to get the table to fit as desired have failed.

2. The CUSTOMER_COUNTRY label is reduced to one word.

3. The CUSTOMER_GROUP label is reduced to one word. The WIDTH= attribute is set to 1.5 inches.

4. The CUSTOMER_AGE_GROUP label is changed to “Age Group”.

5. The width for the ORDER_TYPE variable is set to 1.3 inches.

6. The COSTPRICE_PER_UNIT label is changed to “Cost Price”. The WIDTH= attribute is set to 1.2 inches.

7. The label for TOTAL_RETAIL_PRICE is reduced, and the width is set to 1.2 inches.

8. The PAGE option in the BREAK statement divides the long table into multiple short tables, one for each
country. Each table, one for each country, starts on a new slide.
Output 9.11 Modified Long and Wide Table (Slides 1 and 2)

<table>
<thead>
<tr>
<th>Customer Info</th>
<th>Order Info</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Country</td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
</tr>
</tbody>
</table>
Although Output 9.11 displays only the first two slides, Example 9.11 generates nine slides. That is an improvement of 12 slides. The font size might be smaller, but readability has improved.

Only eight changes were made to the code in Example 9.10 to produce the code in Example 9.11. However, many attempts at creating a desirable slide were required to create the final code. It is not easy to shrink tables for the PowerPoint destination. Producing smaller, attractive tables might take trial and error. Also, it is important to keep in mind that some tables just might not fit on the slide. In those instances, a new strategy for conveying the required information might be preferable.

The PAGE option was used in the BREAK statement in Example 9.11. The PAGE option can also be used in the DEFINE statement. The PAGE option used in the DEFINE statement causes a vertical page break. In other words, it divides sets of columns into separate, smaller tables. The PAGE option in the DEFINE statement is extremely useful for very wide tables.

The PAGE option in the DEFINE statement is also a good strategy for dividing columns in a logical manner. Example 9.12 uses the PAGE option in the DEFINE statement for COSTPRICE_PER_UNIT so that it stays together with TOTAL_RETAIL_PRICE. The result is two tables. The first table contains the customer variables, CUSTOMER_AGE_GROUP and CUSTOMER_GROUP, along with QUANTITY. The second table contains the ID variables and COSTPRICE_PER_UNIT and TOTAL_RETAIL_PRICE. Both tables are shown in Output 9.12.

<table>
<thead>
<tr>
<th>Customer Info</th>
<th>Order Info</th>
</tr>
</thead>
<tbody>
<tr>
<td>Country</td>
<td>Group</td>
</tr>
<tr>
<td>Canada</td>
<td>Internet/Catalog Customers</td>
</tr>
<tr>
<td>Orion Club Gold members</td>
<td>15-30 years</td>
</tr>
<tr>
<td>Orion Club Gold members</td>
<td>31-45 years</td>
</tr>
<tr>
<td>Orion Club Gold members</td>
<td>46-60 years</td>
</tr>
<tr>
<td>Orion Club Gold members</td>
<td>15-30 years</td>
</tr>
<tr>
<td>Orion Club Gold members</td>
<td>31-45 years</td>
</tr>
<tr>
<td>Orion Club Gold members</td>
<td>46-60 years</td>
</tr>
<tr>
<td>Orion Club Gold members</td>
<td>61-75 years</td>
</tr>
<tr>
<td>Orion Club Gold members</td>
<td>31-45 years</td>
</tr>
<tr>
<td>Orion Club Gold members</td>
<td>46-60 years</td>
</tr>
<tr>
<td>Canada</td>
<td>$2,293.80</td>
</tr>
</tbody>
</table>
Example 9.12 Utilizing PAGE Option in the DEFINE Statement

```sas
title;
ods powerpoint file="Output9-12.pptx";
proc report data=orders style(header)=
[backgroundcolor=#3D5AAE color=white];
  column customer_age_group customer_group quantity costprice_per_unit
    total_retail_price;
  define customer_age_group / group style(column)=
[backgroundcolor=#F0F0F0];
  define customer_group / group id  style(column)=
[backgroundcolor=#F0F0F0];
  define quantity / style(column)=
[backgroundcolor=#6EC6F0];
  define costprice_per_unit / page style(column)=
[width=1.4in
    backgroundcolor=#E7ACC6];
  define total_retail_price / "Total Retail Price" style(column)=
[width=1.4in
    backgroundcolor=#E7ACC6];run;
ods powerpoint close;
```

Output 9.12 PAGE Option Generates Two Tables

<table>
<thead>
<tr>
<th>Customer Age Group</th>
<th>Customer Group Name</th>
<th>Quantity Ordered</th>
</tr>
</thead>
<tbody>
<tr>
<td>15-30 years</td>
<td>Orion Club Gold members</td>
<td>132</td>
</tr>
<tr>
<td></td>
<td>Orion Club members</td>
<td>172</td>
</tr>
<tr>
<td>31-45 years</td>
<td>Internet/Catalog Customers</td>
<td>38</td>
</tr>
<tr>
<td></td>
<td>Orion Club Gold members</td>
<td>102</td>
</tr>
<tr>
<td></td>
<td>Orion Club members</td>
<td>257</td>
</tr>
<tr>
<td>46-60 years</td>
<td>Internet/Catalog Customers</td>
<td>39</td>
</tr>
<tr>
<td></td>
<td>Orion Club Gold members</td>
<td>56</td>
</tr>
<tr>
<td></td>
<td>Orion Club members</td>
<td>80</td>
</tr>
<tr>
<td>61-75 years</td>
<td>Internet/Catalog Customers</td>
<td>61</td>
</tr>
<tr>
<td></td>
<td>Orion Club Gold members</td>
<td>18</td>
</tr>
<tr>
<td></td>
<td>Orion Club members</td>
<td>123</td>
</tr>
</tbody>
</table>
9.4.3 Many Header Rows

Spanning headers and headers that are generated by ACROSS variables tend to make many rows of headers. In PDF or HTML, a table with three or four rows of headers is a common practice. However, in PowerPoint, excessive headers rows might not be a good idea. The number of header rows affects the number of data rows that can fit on a slide.

Example 9.13 has an ACROSS variable, ORDER_DATE. The variable QUANTITY becomes almost like an ACROSS variable because three statistics are stacked below it. The result of having these two ACROSS variables is shown in Output 9.13. The header section has four rows of text, but it requires five rows to display because two of the headers wrap within the cell. The table length requires a second slide.

**Example 9.13 Using Two ACROSS Variables**

```plaintext
title;
ods powerpoint file="Output9-13.pptx";
proc report data=orders(obs=70) style(header)=backgroundcolor=#3D5AAE color=white;
   where year(order_date)<2005;
   column ("Customer Info" customer_country customer_age_group)
   order_date, (quantity, sum min max));
```
define customer_country / group format=$cntry.
    style(column)=[backgroundcolor=#F0F0F0];
define customer_age_group / group style(column)=[background=#F0F0F0];
define order_date / across format=year4.;
define sum / style(column)=[backgroundcolor=#FDC88F];
define min / style(column)=[backgroundcolor=#FDC88F];
define max / style(column)=[backgroundcolor=#FDC88F];
run;
ods powerpoint close;

Output 9.13 Header Section Generated by Two ACROSS Variables

<table>
<thead>
<tr>
<th>Customer Country</th>
<th>Customer Age Group</th>
<th>Quantity Ordered</th>
<th>Quantity Ordered</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td></td>
<td>sum   min  max</td>
<td>sum   min  max</td>
</tr>
<tr>
<td>Australia</td>
<td>31-45 years</td>
<td>10    1    4</td>
<td>6     1    2</td>
</tr>
<tr>
<td></td>
<td>61-75 years</td>
<td>4     4    4</td>
<td>.     .    .</td>
</tr>
<tr>
<td>Canada</td>
<td>61-75 years</td>
<td>1     1    1</td>
<td>3     3    3</td>
</tr>
<tr>
<td>Germany</td>
<td>15-30 years</td>
<td>.     .    .</td>
<td>1     1    1</td>
</tr>
<tr>
<td></td>
<td>61-75 years</td>
<td>8     1    3</td>
<td>.     .    .</td>
</tr>
<tr>
<td>Israel</td>
<td>15-30 years</td>
<td>.     .    .</td>
<td>2     1    1</td>
</tr>
<tr>
<td></td>
<td>31-45 years</td>
<td>5     1    2</td>
<td>2     2    2</td>
</tr>
<tr>
<td></td>
<td>46-60 years</td>
<td>4     1    1</td>
<td>5     2    3</td>
</tr>
<tr>
<td>United States</td>
<td>15-30 years</td>
<td>9     1    2</td>
<td>17    1    2</td>
</tr>
</tbody>
</table>
Some changes can be made to limit the number of header rows in an attempt to force the table from Example 9.13 to fit onto one slide. Keep in mind that the acceptable changes vary from table to table and according to business need. For example, the following code removes the word “quantity” from the headers. Without other context, the audience, or reader of the slides, might not know what statistics are used. You must be careful when trying to limit the headers rows. You do not want to lose the meaning of the table. The technique in Example 9.14 illustrates possible changes that could be made.

Example 9.14 demonstrates how to modify the COLUMN statement and labels for the ACROSS variables. Output 9.14 shows a table that fits on one slide.

**Example 9.14 Modifying Labels to Change ACROSS Headers**

```plaintext
options nonumber;
title;
ods powerpoint file="Output9-14.pptx";
proc report data=orders(obs=70)
   style(header)=[backgroundcolor=#3D5AAE color=white];
   where year(order_date)<2005;
   column ("Customer Info"
      customer_country customer_age_group)
   order_date,(quantity, (sum min max));
   define customer_country / group format=$cntry.
   style(column)=[backgroundcolor=#F0F0F0];
```

<table>
<thead>
<tr>
<th>Customer Info</th>
<th>Quantity Ordered</th>
<th>Quantity Ordered</th>
</tr>
</thead>
<tbody>
<tr>
<td>Customer Age Group</td>
<td>sum</td>
<td>min</td>
</tr>
<tr>
<td>31-45 years</td>
<td>20</td>
<td>1</td>
</tr>
<tr>
<td>46-60 years</td>
<td>10</td>
<td>1</td>
</tr>
<tr>
<td>61-75 years</td>
<td>7</td>
<td>2</td>
</tr>
</tbody>
</table>
define customer_age_group / group style(column)=[backgroundcolor=#F0F0F0];
define order_date / across format=year4. ''; ❷
define quantity / ''; ❸
define sum / style(column)=[backgroundcolor=#FDC88F];
define min / style(column)=[backgroundcolor=#FDC88F];
define max / style(column)=[backgroundcolor=#FDC88F];
run;
ods powerpoint close;

❶ A blank spanning header is added so that the “Customer Info” spanning header is moved up next to the year value. The blank spanning header ensures that the header row that contained the QUANTITY label is suppressed.

❷ The ORDER_DATE variable’s label is set to missing so that the header row is suppressed.

❸ The QUANTITY variable’s label is set to missing so that the header row is suppressed.

Output 9.14 Modified ACROSS Labels Change Header Section

<table>
<thead>
<tr>
<th>Customer Info</th>
<th>2003</th>
<th>2004</th>
</tr>
</thead>
<tbody>
<tr>
<td>Customer Country</td>
<td>Customer Age Group</td>
<td>sum</td>
</tr>
<tr>
<td>Australia</td>
<td>31-45 years</td>
<td>10</td>
</tr>
<tr>
<td></td>
<td>61-75 years</td>
<td>4</td>
</tr>
<tr>
<td>Canada</td>
<td>61-75 years</td>
<td>1</td>
</tr>
<tr>
<td>Germany</td>
<td>15-30 years</td>
<td>.</td>
</tr>
<tr>
<td></td>
<td>61-75 years</td>
<td>8</td>
</tr>
<tr>
<td>Israel</td>
<td>15-30 years</td>
<td>.</td>
</tr>
<tr>
<td></td>
<td>31-45 years</td>
<td>5</td>
</tr>
<tr>
<td></td>
<td>46-60 years</td>
<td>4</td>
</tr>
<tr>
<td>United States</td>
<td>15-30 years</td>
<td>9</td>
</tr>
<tr>
<td></td>
<td>31-45 years</td>
<td>20</td>
</tr>
<tr>
<td></td>
<td>46-60 years</td>
<td>10</td>
</tr>
<tr>
<td></td>
<td>61-75 years</td>
<td>7</td>
</tr>
</tbody>
</table>

Even with the modifications made in Example 9.14, the table barely fits on the slide in Output 9.14. Also, the slide does not contain the slide number. The entire table would not fit on the slide if the slide number was included.
The examples in this chapter have limited the number of observations from the input data set. Also, these examples do not have titles or footnotes. Hopefully, this demonstrates that even when using a small number of observations without titles or footnotes, and sometimes without a slide number, it is difficult to prevent a table from wrapping to a second or third slide. Even more adjustments might be required to fit tables on slides when multiple titles are used. Often, because company logos are placed on slides, they must be taken into account when altering tables to fit on slides.

Trial and error is required to get tables to fit into a presentation and in a pleasing manner. Changes such as removing header rows or breaking the table apart are techniques that help achieve the desired appearance. Each presentation is different. Remember to set and maintain reasonable expectations.

The ODS destination for PowerPoint also allows the use of ODS LAYOUT. ODS LAYOUT allows multiple objects such as tables and graphs to be placed on a single slide. If a table does not fit inside a layout container, it will be suppressed. Keeping a table small and compact is vital when placing it inside a layout container for PowerPoint.

### 9.5 Inline Formating Functions

PROC REPORT tables often contain extra formatting techniques (such as colors, images, dollar signs, and Unicode values) to enhance the display of information. Trafficlighting and numeric formatting work the same way in the PowerPoint destination as they do in all other destinations for PROC REPORT. Images require a specific image attribute, which is discussed in Section 9.6.

Inline formatting is a popular way of adding features to the report, but not all functions are supported for the destination. The PowerPoint destination supports these inline functions:

- dagger
- nbspace
- newline
- sigma
- style
- sub
- super
- unicode

This list covers most of the needs of PROC REPORT output. However, it is important to note the inline functions that are not supported by the destination. Most of the functions that are not supported are generally used outside tables and are more likely found in TITLE and FOOTNOTE statements. Also, some of the functions are specific to other ODS destinations. The PowerPoint destination does not support these functions:

- date
- dest
- lastpage
- leaders
Also, some cell attributes, used with inline formatting, cannot be changed. They include the following:

- background of a shape (for example, the color of a word)
- pre-text and post-text
- alignment
- first-line text indentation

You might be able to apply the attribute using a CALL DEFINE statement. However, you will need to test the attribute. Example 9.15 creates a new variable within a DATA step that has inline formatting and that includes POSTTEXT=. The asterisk does not appear in the column for C_G in the table, Output 9.15. The asterisk does appear for CUSTOMER_GENDER, which used a CALL DEFINE statement to apply the POSTTEXT= attribute.

**Example 9.15 Using POSTTEXT=**

```sas
ods escapechar="~";
data orders2;
   set orders;
   length c_g $50;
   c_g = catt('~{style [posttext="*"],customer_gender,'}');
run;
ods powerpoint file="Output9-15.pptx" dpi=300;
proc report data=orders2 style(header)=[backgroundcolor=#CCDF9C];
   column c_g customer_gender n;
   define c_g / group;
   define customer_gender / group;
   compute customer_gender;
      if customer_gender='F' then
         call define(_col_,'style','style=[posttext="*"']);
      if customer_gender='F' then
         call define(_row_,'style','style=[color=#8E2F8A]');
      else call define(_row_,'style','style=[color=#3D5AAE]');
   endcomp;
run;
ods powerpoint close;
```
The workaround for the first line indention functionality is to use the supported NBSPACE and NEWLINE functions. Example 9.17 demonstrates how to concatenate three variables together, add a newline, and add a space before the second and third values. The results are shown in Output 9.17.

**Example 9.16 Using the NEWLINE and NBSPACE Functions**

```plaintext
ods escapechar="^";
ods powerpoint file="Output9-16.pptx";
proc report data=orders(obs=5) style(header)=[backgroundcolor=#62BAE9];
  column customer_name customer_gender customer_age customer;
  define customer_age / display;
  define customer / computed "Customer Info"
    style(column)=[backgroundcolor=#FFE7A1];
  compute customer / char length=100;
    customer = catx("^{newline 1}^{nbspace 5},
      customer_name,customer_gender,customer_age);
  endcomp;
run;
ods powerpoint close;
```
### Output 9.16 Gender and Age Appear on Separate Rows and Are Indented

<table>
<thead>
<tr>
<th>Customer Name</th>
<th>Customer Gender</th>
<th>Customer Age</th>
<th>Customer Info</th>
</tr>
</thead>
<tbody>
<tr>
<td>Kyndal Hooks</td>
<td>F</td>
<td>43</td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>Kyndal Hooks</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>F</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>43</td>
</tr>
<tr>
<td>Annmarie Leveille</td>
<td>F</td>
<td>23</td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>Annmarie Leveille</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>F</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>23</td>
</tr>
<tr>
<td>Najma Hicks</td>
<td>F</td>
<td>21</td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>Najma Hicks</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>F</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>21</td>
</tr>
<tr>
<td>Yan Kozlowski</td>
<td>M</td>
<td>38</td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>Yan Kozlowski</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>M</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>38</td>
</tr>
<tr>
<td>Kyndal Hooks</td>
<td>F</td>
<td>43</td>
<td></td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>Kyndal Hooks</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>F</td>
</tr>
<tr>
<td></td>
<td></td>
<td></td>
<td>43</td>
</tr>
</tbody>
</table>

### 9.6 Style Attributes for Images

Images are important for most presentations. They attract the audience’s attention and provide useful information. Images can be added above and below a PROC REPORT table with the PREIMAGE= and POSTIMAGE= attributes, respectively. However, these attributes cannot be used to include images inside table cells. Microsoft PowerPoint does not allow pre- and postimages. Only background images can be used inside table cells. In other words, the BACKGROUNDIMAGE= attribute can be used for a table cell, but the PREIMAGE= or POSTIMAGE= attributes cannot be used for a table cell. By allowing background images, the PowerPoint destination is similar to HTML, but is different from PDF or RTF.

Example 9.17 places an image above and below the table using the PREIMAGE= and POSTIMAGE= attributes within the STYLE(REPORT) location. The results are shown in Output 9.17.

**Example 9.17 Using PREIMAGE= and POSTIMAGE=**

```plaintext
ods powerpoint file="Output9-17.pptx";
proc report data=orders
    style(header)=[backgroundcolor=#E7ACC6]
    style(report)=[preimage="cologo.tif" postimage="cologo.tif"];
```
column customer_type n;
  define customer_type / group;
run;
ods powerpoint close;

Output 9.17 Images Placed Above and Below the Table

Example 9.18 places images inside the table cells. The BACKGROUNDIMAGE= attribute must be used. Using a different image attribute will result in a blank cell. Output 9.18 shows the results.

Example 9.18 Using BACKGROUNDIMAGE=

ods powerpoint file="Output9-18.pptx";
proc report data=orders style(header)=[backgroundcolor=#62BAE9];
column order_type;
define order_type / group format=typef.
  style(column)=[textalign=c width=2in height=2in];
compute order_type;
  if order_type = 1 then call define(_col_,'style',
    'style=[backgroundimage="goldmedal.tiff"]');
  else if order_type = 2 then call define(_col_,'style',
    'style=[backgroundimage="silvermedal.tiff"]');
  else if order_type = 3 then call define(_col_,'style',
    'style=[backgroundimage="bronzemedal.tiff"]');
endcomp;
run;
ods powerpoint close;
When working with images for this destination, it is important to understand that by default the image will be stretched or shrunk to fit the cell. The destination ensures that the image fits inside the box that encloses it.

The entire slide can have a background image, as well. Example 9.19 demonstrates how to include a background image for the slide, which is specified in the ODS POWERPOINT statement. Because no other suboptions are specified, the image is stretched to fit the full slide. A background color is set for the columns in the table so that the table values are still readable, as shown in Output 9.19.

**Example 9.19 Use BACKGROUNDIMAGE= for the Slide**

```sas
ods powerpoint file="Output9-19.pptx"
   options(backgroundimage="saslogol.png");
proc report data=orders
   style(header)=[backgroundcolor=#62BAE9]
   style(column)=[backgroundcolor=#F0F0F0];
   column customer_type n;
   define customer_type / group;
run;
ods powerpoint close;
```
The background image in Output 9.19 is **pixelated** because it has been stretched. There are two methods for preventing this behavior. The first is to create an image that is the size of the slide. The second option is to use the BACKGROUNDREPEAT= suboption.

Example 9.20 demonstrates how to use an image that is the size of the slide. Output 9.20 shows the image in an appropriate size.

**Example 9.20 Use an Image That Is the Size of the Slide**

```plaintext
ods powerpoint file="Output9-20.pptx"
  options(backgroundimage="saslogo_pptx.png");
proc report data=orders
  style(header)=[backgroundcolor=#62BAE9]
  style(column)=[backgroundcolor=#F0F0F0];
  column customer_type n;
  define customer_type / group;
run;
ods powerpoint close;
```
Output 9.20 Image Fits on the Slide and Is Not Stretched

<table>
<thead>
<tr>
<th>Customer Type Name</th>
<th>n</th>
</tr>
</thead>
<tbody>
<tr>
<td>Internet/Catalog Customers</td>
<td>76</td>
</tr>
<tr>
<td>Orion Club members high activity</td>
<td>115</td>
</tr>
<tr>
<td>Orion Club members medium activity</td>
<td>154</td>
</tr>
<tr>
<td>Orion Club Gold members high activity</td>
<td>65</td>
</tr>
<tr>
<td>Orion Club Gold members low activity</td>
<td>42</td>
</tr>
<tr>
<td>Orion Club Gold members medium activity</td>
<td>77</td>
</tr>
<tr>
<td>Orion Club members low activity</td>
<td>88</td>
</tr>
</tbody>
</table>

The image will retain its original size if the BACKGROUNDREPEAT= suboption is set to REPEAT, REPEAT_X, or REPEAT_Y. Example 9.21 demonstrates how to use the original, smaller image along with the BACKGROUNDREPEAT= suboption. Output 9.21 shows the effect of the suboption.

Example 9.21 Using BACKGROUNDREPEAT=

```sas
ods powerpoint file="Output9-21.pptx"
   options(backgroundimage="saslogo1.png" backgroundrepeat="repeat");
proc report data=orders
   style(header)=[bgcolor=#62BAE9]
   style(column)=[bgcolor=#F0F0F0];
   column customer_type n;
   define customer_type / group;
run;
ods powerpoint close;
```
Output 9.21 Image Is Repeated

<table>
<thead>
<tr>
<th>Customer Type Name</th>
<th>n</th>
</tr>
</thead>
<tbody>
<tr>
<td>Internet/Catalog Customers</td>
<td>76</td>
</tr>
<tr>
<td>Orion Club members high activity</td>
<td>115</td>
</tr>
<tr>
<td>Orion Club members medium activity</td>
<td>154</td>
</tr>
<tr>
<td>Orion Club Gold members high activity</td>
<td>65</td>
</tr>
<tr>
<td>Orion Club Gold members low activity</td>
<td>42</td>
</tr>
<tr>
<td>Orion Club Gold members medium activity</td>
<td>77</td>
</tr>
<tr>
<td>Orion Club members low activity</td>
<td>88</td>
</tr>
</tbody>
</table>

9.7 Key Takeaways

This chapter provided a lot of information about handling the unique aspects of the ODS destination for PowerPoint. The beginning of the chapter discussed the changes that can be made with the style template. The important takeaways regarding the style template are the following:

- Always use PowerPointLight or PowerPointDark as the parent style when creating a custom style template for PowerPoint.
- Make changes to the style elements only when those changes are needed for every table in the presentation.
- BORDERSTYLE=HIDDEN removes the gray border lines inside the table.
This chapter also covered programming techniques to help ensure that the PROC REPORT table displays as desired in the PowerPoint slide. Key takeaways include the following:

- Limit the number of columns and rows in a table as much as possible.
- Keep the number of header rows to a minimum.
- Use the PAGE option in the DEFINE and BREAK statements to divide large tables into smaller tables.
Ready to take your SAS® and JMP® skills up a notch?

Be among the first to know about new books, special events, and exclusive discounts.

support.sas.com/newbooks

Share your expertise. Write a book with SAS.

support.sas.com/publish

sas.com/books for additional books and resources.